```
(* This file verifies a number of calculations used in the paper
 titled "Option Value of Cash," by Jialin Yu (Columbia Business School) *)
```

```
(* Note: Gamma distribution[a,b] in the paper
  is equivalent to GammaDistribution[a,1/b] in Mathematica *)
(* probability density function of Gamma prior *)
Simplify[PDF[GammaDistribution[a, 1 / b], λ], b > 0 && a > 0 && λ > 0]
```

$$\frac{b^a\, e^{-b\,\lambda}\, \lambda^{-1+a}}{\text{Gamma}[a]}$$

```
(******************** verify Lemma 1 on Bayesian Posterior ********************)
```

$$\text{Simplify}\Big[\text{PDF}\Big[\text{GammaDistribution}\Big[a, \frac{1}{b+\Delta}\Big], \lambda\Big] - \frac{\text{PDF}\big[\text{GammaDistribution}\big[a, \frac{1}{b}\big], \lambda\big]\, e^{-\lambda\,\Delta}}{\int_0^\infty \text{PDF}\big[\text{GammaDistribution}\big[a, \frac{1}{b}\big], \lambda\big]\, e^{-\lambda\,\Delta}\, d\lambda},\ a > 0\ \&\&\ b > 0\ \&\&\ \Delta > 0\Big]$$

$$0$$

```
(******************** verify Lemma 2 ***************************)
(* posterior mean and variance *)
Mean[GammaDistribution[a, 1 / b]]
Variance[GammaDistribution[a, 1 / b]]
(* expected downturn length *)
```

$$\text{FullSimplify}\Big[\int_0^\infty \frac{1}{\lambda}\, \text{PDF}[\text{GammaDistribution}[a, 1/b], \lambda]\, d\lambda,\ a > 1\ \&\&\ b > 0\Big]$$

$$\frac{a}{b}$$

$$\frac{a}{b^2}$$

$$\frac{b}{-1+a}$$

```
(****************** formula for NPV(λ) ************************)
```

$$\text{Simplify}\Big[\int_0^\infty e^{-r\,t}\, D\, dt - \int_0^\infty \Big(\int_0^t e^{-r\,s}\, \delta\, ds\Big)\, \lambda\, e^{-\lambda\,t}\, dt,\ r > 0\ \&\&\ \lambda > 0\Big]$$

$$\frac{D}{r} - \frac{\delta}{r+\lambda}$$

```
(****************** formula for V(a,b) ************************)
```

$$V = \int_0^{\infty} \left(\frac{D}{r} - \frac{\delta}{r + \lambda}\right) \text{PDF}[\text{GammaDistribution}[a, 1/b], \lambda]\, d\lambda;$$

$$\text{FullSimplify}\left[V - \left(\frac{D}{r} - \delta\, b^{a}\, e^{r\, b}\, r^{a-1}\, \text{Gamma}[1 - a, r\, b]\right), a > 0\ \&\&\ b > 0\ \&\&\ r > 0\right]$$

0

```
(*******************************************************************)
(* verify the statement in the proof of proposition
  1 that the limit of b^a e^(r b) r^(a-1) Gamma[1-a,r b] is 1/r when b→∞ *)
```

$$\text{Limit}\left[b^{a}\, e^{r\, b}\, r^{a-1}\, \text{Gamma}[1 - a, r\, b], b \to \infty\right]$$

$$\frac{1}{r}$$

```
(************** Solve for j in the proof of proposition 3 ***********)
```

$$\text{Collect}\Big[\text{Solve}\Big[\text{Simplify}\Big[\text{Eliminate}\Big[\Big\{\frac{b + n}{a - 1} == \frac{b_L + i\, g + n}{a_L - 1}, \frac{a_L}{b_L + i\, g + n + t} == \frac{a}{b + n + t}, \frac{b + n + t}{a - 1} == \frac{b_L + j\, g + n + t}{a_L - 1}\Big\}, \{n, t\}\Big], a > 1\ \&\&\ a_L > 1\Big], j\Big], i\Big]$$

$$\left\{\left\{j \to \frac{a\, i}{-1 + a} + \frac{-b + b_L}{(-1 + a)\, g}\right\}\right\}$$

```
(*************************************************************************)
(* Solve for t1*, t2* and n in the proof of proposition 4 *)
(* First, t1* and n *)
```

$$\text{Collect}\Big[\text{Simplify}\Big[\text{Solve}\Big[\Big\{\frac{b + n}{a - 1} == \frac{b_L + i\, g + n}{a_L - 1}, \frac{a_L}{b_L + i\, g + n + t} == \frac{a}{b + n + t}\Big\}, \{n, t\}\Big]\Big], \{g, i, b\}\Big]$$

```
(* Next, t2* *)
```

$$\text{Solve}\Big[\text{Simplify}\Big[\text{Eliminate}\Big[\Big\{\frac{b + n}{a - 1} == \frac{b_L + i\, g + n}{a_L - 1}, \frac{b + n + t}{a - 1} == \frac{b_L + j\, g + n + t}{a_L - 1}\Big\}, \{n\}\Big], a > 1\ \&\&\ a_L > 1\Big], t\Big] \,/.\, j \to \frac{K}{M}$$

$$\left\{\left\{t \to \frac{b}{a - a_L} - \frac{g\, i}{a - a_L} - \frac{b_L}{a - a_L}, n \to -\frac{(-1 + a)\, g\, i}{a - a_L} - \frac{b\, (1 - a_L)}{a - a_L} - \frac{(-1 + a)\, b_L}{a - a_L}\right\}\right\}$$

$$\left\{\left\{t \to \frac{(-1 + a)\, g\, \left(i - \frac{K}{M}\right)}{a - a_L}\right\}\right\}$$

```
(******************************* calculations in Figure 2 *****************************)
(* define posterior valuation formula*)
```

$$V[D\_, \delta\_, r\_, a\_, b\_] := \frac{D}{r} - \delta\, b^{a}\, e^{b\, r}\, r^{-1+a}\, \mathrm{Re}[\mathrm{Gamma}[1-a, b\, r]]$$

```
(* pessimist discount at beginning of downturn *)
```

$$p = V\left[1, \frac{9}{10}, 0.005, \frac{9}{8}, \frac{3}{4}\right]$$

$$o = V\left[1, \frac{9}{10}, 0.005, 2, 1\right]$$

$$p / o - 1$$

```
(* discount after 2 years *)
```

$$p = V\left[1, \frac{9}{10}, 0.005, \frac{9}{8}, \frac{3}{4} + 24\right]$$

$$o = V\left[1, \frac{9}{10}, 0.005, 2, 1 + 24\right]$$

$$p / o - 1$$

197.515

199.121

-0.00806784

164.102

182.674

-0.101667

```
(**************************** calculations in Figure 3 ********************************)
(* define posterior valuation formula*)
```

$$V[D\_, \delta\_, r\_, a\_, b\_] := \frac{D}{r} - \delta\, b^{a}\, e^{b\, r}\, r^{-1+a}\, \mathrm{Re}[\mathrm{Gamma}[1-a, b\, r]]$$

```
(* the valuations of the optimists and pessimist i=0 at the beginning of downturn *)
```

$$V\left[1, \frac{9}{10}, 0.005, \frac{26}{25}, 1\right]$$

$$V\left[1, \frac{9}{10}, 0.005, 3, 9\right]$$

```
(* time when the buy-and-hold valuation of i=0 equals that of the optimists *)
```

$$\mathrm{FindRoot}\left[V\left[1, \frac{9}{10}, 0.005, 3, 9 + \tau\right] == V\left[1, \frac{9}{10}, 0.005, \frac{26}{25}, 1 + \tau\right], \{\tau, 0\}\right]$$

```
(* time when i=0 steps in, solved from Proposition 2 *)
entry0 = τ /. FindRoot[3/(9 + τ) == (26/25)/(1 + τ), {τ, 0}]
(* valuations of the optimists and pessimist i=0 when i=0 steps in *)
V[1, 9/10, 0.005, 26/25, 1 + entry0]
V[1, 9/10, 0.005, 3, 9 + entry0]
(* solve for the time the last liquidity provider i=K/M steps in *)
K = 1; M = 5; g = 500;
entry1 = τ /. FindRoot[V[1, 9/10, 0.005, 3, 9 + g K/M + τ] == V[1, 9/10, 0.005, 26/25, 1 + τ], {τ, 0}]
(* solve for those investors who buy at the same time as i=
 K/M. The result shows that investors i≥0.084 buy together with i=K/M *)
FindRoot[3/(9 + g i + entry1) == (26/25)/(1 + entry1), {i, 0}]
(* time when the buy-and-hold valuation of i=0.1 matches the market price *)
entry2 = τ /. FindRoot[V[1, 9/10, 0.005, 3, 9 + g 0.1 + τ] == V[1, 9/10, 0.005, 26/25, 1 + τ], {τ, 0}]
(* the valuation of i=0.1 when his buy-and-hold valuation matches the market price *)
V[1, 9/10, 0.005, 3, 9 + g 0.1 + entry2]
(* the market price and the valuation of i=0.1 when he enters *)
V[1, 9/10, 0.005, 26/25, 1 + entry1]
V[1, 9/10, 0.005, 3, 9 + g 0.1 + entry1]
(* solve for the total fraction liquidated 1 year and 2 year into the downturn *)
M * f /. FindRoot[3/(9 + g f + 12) == (26/25)/(1 + 12), {f, 0}]
M * f /. FindRoot[3/(9 + g f + 24) == (26/25)/(1 + 24), {f, 0}]
(* total fraction liquidated 1 year and 2 year
 into the downturn if there were no option value to delay *)
M * f /. FindRoot[V[1, 9/10, 0.005, 3, 9 + g f + 12] == V[1, 9/10, 0.005, 26/25, 1 + 12], {f, 0}]
M * f /. FindRoot[V[1, 9/10, 0.005, 3, 9 + g f + 24] == V[1, 9/10, 0.005, 26/25, 1 + 24], {f, 0}]
```

```
196.129

196.11

{τ → 0.00693863}

3.2449

188.049

194.777
```

25.4538

{i → 0.0837105}

10.9442

175.192

159.062

171.155

0.165

0.391154

0.539781

0.953331